FPN 13

This work was supported by the United States Atomic Energy Commission.
Contract Number AT-(29-1)-789

SCL-DR-69-44

NUMERICAL PREDICTIONS OF THE MOTION OF MAGNETICALLY ACCELERATED FLYER PLATES

R. W. Reynolds
Simulation Division 8343
R. S. Jacobson
Experimental Mechanics Division 8122
Sandia Laboratories, Livermore

July 1969

ABSTRACT

A numerical solution predicts the motion of a magnetically accelerated flat flyer plate. Results obtained with the solution agree closely with experimental data.

This work was supported by the United States Atomic Energy Commission. Contract Number AT-(29-1)-789.

ACKNOWLEDGMENT

The authors wish to thank K. D. Marx, Analytical Division 8341, for his assistance in the theoretical portion of this report, W. B. Vandermolen, Experimental Mechanics Division and R. E. Johnson, Simulation Division 8343, who conducted the flyer experiments.

CONTENTS

ILLUSTRATIONS

NUMERICAL PREDICTIONS OF THE MOTION OF MAGNETICALLY ACCELERATED FLYER PLATES

Introduction

Capacitor banks have been successfully used to accelerate flyer plates for impulse testing. The predictions of the motion of the flyer plate should be of considerable assistance in the design of capacitor banks and flyer plates.

In this report, the motion of the flyer plate is studied with the aid of a circuit equation and an equation of motion. These two equations are derived in general, applied to a flat flyer geometry, and then solved numerically using a Runge-Kutta technique. A series of experimental tests was conducted with a capacitor bank at Sandia Laboratories, Livermore; the results of these tests were in very good agreement with the numerical predictions.

Theory

The model used to obtain the circuit equation for the flyer-bank system is shown in Figure 1. The flyer and its return path are represented by an inductance L_F, which is a function of the separation of the flyer and its return path X, and a constant resistance R_F. If the total inductance is L and the total resistance is R, L_B and R_B can be defined such that

$$L_B = L(x) - L_F(x) \tag{1}$$

$$R_B = R - R_F \tag{2}$$

where L_B and R_B can be considered to be the total remaining inductance and resistance.



Figure 1. Schematic Representation of Flyer Plate and Capacitor Bank

The circuit equation can be written directly as follows

$$\frac{d}{dt}(LI) + RI + \int_0^t \frac{I}{C} dt = V_0 \qquad (3)$$

where V_0 is the initial voltage on the capacitor C, I is the current flowing in the circuit, and t is time. Since the switch is closed at t = 0,

$$I(0) = 0 \ . \qquad (4)$$

By differentiation,

$$\frac{d^2}{dt^2}(LI) + R\frac{dI}{dt} + \frac{I}{C} = 0 \qquad (5)$$

where

$$\frac{d}{dt}(LI)\Big|_{t=0} = V_0 \qquad (6)$$

The equation of motion can be solved from virtual energy considerations. Considering the inputs to the flyer plate section of the equivalent circuit, energy conservation requires that

$$V_F I \delta t = R_F I^2 \delta t + \delta\left(\frac{1}{2} L_F I^2\right) + f \delta x \tag{7}$$

where

V_F	= voltage across flyer equivalent circuit
f	= force acting on the flyer
$V_F I \delta t$	= total energy into the flyer equivalent circuit
$R_F I^2 \delta t$	= energy lost by resistive heating
$\delta\left(\frac{1}{2} L_F I^2\right)$	= change in energy stored in the magnetic field
$f \delta x$	= mechanical work done in moving the flyer plate

From circuit considerations,

$$V_F = \frac{d\phi}{dt} + R_F I \tag{8}$$

where ϕ, the magnetic flux, $= L_F I$. Equation (8) may be expressed in virtual form as

$$V_F \delta t = \delta\left(L_F I\right) + R_F I \delta t \tag{9}$$

Using this result Equation (7) can be rewritten as

$$\left[\delta\left(L_F I\right) + R_F I \delta t\right] I = R_F I^2 \delta t + \delta\left(\frac{1}{2} L_F I^2\right) + f \delta x, \tag{10}$$

thereby obtaining

$$f \delta x = \frac{1}{2} I^2 \delta L_F$$

or

$$f = \frac{1}{2} I^2 \frac{dL_F}{dx} \tag{11}$$

Knowing $f = m \frac{d^2x}{dt^2}$, where m = flyer mass, we obtain

$$\frac{d^2x}{dt^2} = \frac{I^2}{2m} \frac{dL_F}{dx} . \tag{12}$$

Since the initial displacement is x_0 and the initial velocity is zero,

$$x(0) = x_0 \tag{13}$$

$$\left.\frac{dx}{dt}\right|_{t=0} = 0 \tag{14}$$

The inductance of the flyer can be expressed as

$$L_F(x) = \int_{x_0}^{x} \left(\frac{dL_F}{dx}\right) dx + L_F\left(x_0\right) \tag{15}$$

where $L_F(x_0)$ is the inductance of the flyer at its initial distance x_0.

Thus, Equation (1) becomes

$$L(x) = L_B + \int_{x_0}^{x} \left(\frac{dL_F}{dx}\right) dx + L_F\left(x_0\right) \tag{16}$$

or

$$L(x) = L_B + \int_{0}^{x} \left(\frac{dL_F}{dx}\right) dx$$

At this point, Equations (5), (12), and (16) are sufficient to predict flyer plate characteristics. These equations can be applied to any one dimensional geometry if $\frac{dL_F}{dx}$ can be calculated or measured.

Application of Theory to Flat Flyer Geometry

Since much of the flyer testing at Sandia Laboratories, Livermore has been done with a flat-flyer geometry, the remainder of this report will deal with that geometry. The analysis to be discussed can easily be extended to other geometries.

The value of $\frac{dL_F}{dx}$ for the flat flyer can be calculated by first obtaining the magnetic flux density between the flyer plate and the return plate. Using Ampere's Law, it can be determined that

$$B = \frac{\mu I}{w} \tag{17}$$

where w is the width of the plates and μ is the permeability of free space.



Figure 2. Actual Flat Flyer Fixture

The flyer inductance is then given by

$$L_F = \frac{\int_0^x \int_0^{\ell} Bdxdy}{I} = \frac{\mu \ell x}{w}$$

where ℓ is the length of the plates. Differentiating this expression gives

$$\frac{dL_F}{dx} = \frac{\mu \ell}{w} , \tag{18}$$

which is independent of x.

Since this simple analysis neglects fringing and skin effects, $\frac{dL_F}{dx}$ was measured experimentally by mounting the flyer in its holder and using an inductance bridge to measure the inductance as a function of displacement. In this procedure, the flyer was manually moved away from the return plate, and L_F was measured at each new position. To include an estimate of the inductive skin effect, the bridge frequency was set at the ring frequency of the bank.

The results of these measurements are shown in Figure 3. In this particular case, a straight line fit yields $\frac{dL_F}{dx} = 5.12 \mu h/m$, while Equation (18) gives $\frac{dL_F}{dx} = 6.40 \mu h/m$. Equation (16) then becomes

$$L(x) = L_B + \kappa x \tag{19}$$

where K is a constant and can be determined from Equation (18) or Figure 3.

By eliminating L in Equation (5) and $\frac{dL_F}{dx}$ in Equation (12), two differential equations are left:



Figure 3. Flyer Inductance as a Function of Distance

$$\frac{d^2}{dt^2}\left[\left(L_B + \kappa x\right)I\right] + R\frac{dI}{dt} + \frac{I}{C} = 0 \quad (20)$$

$$\frac{d^2x}{dt^2} = \frac{I^2}{2\mu}\kappa \quad (21)$$

subject to the boundary conditions

$$\frac{d}{dt}\left[\left(L_B + \kappa x\right)I\right]\Big|_{t=0} = V_0 \quad (22)$$

$$I(0) = 0 \quad (23)$$

$$\frac{dx}{dt}\Big|_{t=0} = 0 \quad (24)$$

and

$$x(0) = x_0 \quad (25)$$

Numerical Solution of the Differential Equations

The two linked, second-order differential equations (20 and 21) can now be solved numerically. These two second-order equations were reduced to four first-order equations with the independent variables I, x, $\frac{dI}{dt}$, and $\frac{dx}{dt}$. A program was then written for the CDC 6600 computer to numerically solve these four equations. The heart of the program was a modified Runge-Kutta method which is described by Gill in Reference 1.

The program, given in the Appendix, was designed to run a given flyer plate and bank at four different initial voltages. The resulting currents, displacements, and velocities were calculated as a function of time and compared with experimental values.

Computer Predictions and Experimental Results

To check the accuracy of the program, the capacitor bank at Sandia Laboratories, Livermore, was fired at four different voltages using four, identical, flat, aluminum flyer plates. The parameters of the system were:

R	$= 7.4\ m\Omega$	Total resistance
C	$= 108.5\ \mu f$	Total capacitance
$L_B + \kappa x_0$	$= 101.5$ nh	Total initial inductance
V_0	= 5, 9, 12, 16 Kv	Initial bank voltages
ρ	= 2.71 grams/cc	Flyer density
ℓ	= 15.0 cm	Flyer length
w	= 2.94 cm	Flyer width
th	= .0313 cm	Flyer thickness
x_0	= .0254 cm	Initial displacement of flyer

A record of flyer displacement versus time was determined with the aid of a streaking camera. The velocity of the flyer was determined from the slope of these records, and the current information was determined from photographs of oscilloscope traces. For more information concerning the experimental procedure, see Reference 2.

The computer program was inputted with the above values in mks units. The program also requires T_{max} (the maximum time calculated), X_{lim} (the maximum displacement plotted), and $\frac{dL_F}{dx}$. These parameters were input as:

$$T_{max} = 5.0 \times 10^{-5} \text{ seconds}$$

$$X_{lim} = 3.0 \times 10^{-3} \text{ meters}$$

$$\frac{dL_F}{dx} = \kappa = 5.12 \times 10^{-6} \text{ henries/meter}$$

This value of $\frac{dL_F}{dx}$ was obtained experimentally. The computer program was then rerun with $\frac{dL_F}{dx} = 6.40 \times 10^{-6}$ henries/meter, which was calculated from Equation (18).

The results of these two computer runs are shown in Figures 4, 5, and 6. The results from the four experimental shots are included in these plots as data points. The displacement shown in these graphs was measured from the initial position of the flyer, x_0.

Figure 4 shows a plot of flyer velocity versus displacement for four voltages based upon the experimentally determined $\frac{dL_F}{dx}$. The experimental values of velocity are within ten percent of the predicted values. Since the accuracy of experimental measurements was estimated to be only ten percent, these results are considered satisfactory.

Figure 5 is similar to Figure 4, with the exception that the $\frac{dL_F}{dx}$ was used in the computer run. Calculated velocities were all higher than those obtained experimentally, the worst result being high by 15 percent. This is the natural result of using too high a value of $\frac{dL_F}{dx}$ (see Equation 11).

The last graph shows a theoretical plot of the current and displacement curves based on the measured $\frac{dL_F}{dx}$. The experimental points which correspond to the same initial voltage are also plotted on this graph and show close agreement. The unevenness of the experimental plots gives an indication of the error in the data reduction process.



Figure 4. Flyer Velocity Versus Displacement



Figure 5. Flyer Velocity Versus Displacement



Figure 6. Flyer Displacement and Total Current Against a Common Time Axis

Conclusion

The numerical results obtained using the mathematical model of the flat flyer described in this report correlate well with experimental data. These results show that for the most accurate predictions, the value of $\frac{dL_F}{dx}$ should be measured, although fair predictions of flyer motion can be made using the calculated value. It is suspected that for greater width-to-spacing ratios, as would exist with larger flyers, the calculated value of $\frac{dL_F}{dx}$ would correspond more closely to the measured value.

This program has been used to investigate the effects of varying flyer plate and bank parameters. The results of these studies (Reference 3) are useful in flyer plate and bank design.

REFERENCES

1. Gill, S., "A Process of the Step-by-Step Integration of Differential Equations in an Automatic Digital Computing Machine", Proceedings of Cambridge Philosphical Society, Vol. 47, No. 1, 1951, pp 96-108.

2. Jacobson, R. S., Magnetic Acceleration of Flyer Plates for Shock Wave Testing of Materials, SCL-DR-67-60, September 1967.

3. Nelson, D. B., Resistance Effects on the Performance of Magnetically Driven Flyer Plates, SCL-DR-69-34, May 1969.

APPENDIX -- PROGRAM

```
      PROGRAM MOTION (INPUT,OUTPUT,TAPE99)
      COMMON /OUT/ TMAT(300),XMAT(300,4),CMAT(300,4),PMAT(300,4),V(4)
      COMMON /VAR/ R,CAP,K1,K2,K3,  L,XN,AREA,M,MU,PI
      REAL L,K1,K2,K3
      REAL MU,M
      COMMON /NUTS/ H,HMIN,HMAX,XLIML,INTCON,N,TMAX,DELPR,XLIMR,NPL
C
C
C
C        THERE ARE FOUR DATA CARDS
C        ALL VARIABLES ARE IN MKS UNITS
C        IF DL/DX = 0. PROGRAM CALCULATES VALUE
C        ALL VARIABLES REQUIRE 10 SPACES
C        ALL VARIABLES USE E FORMATS EXCEPT TYPE VARIABLE ON SECOND CARD
C        THE TYPE VARIABLE DESCRIBES THE TYPE OF FLYER MATERIAL (H FORMAT)
C        FIRST CARD  $$$$  THE FOUR DIFFERENT BANK VOLTAGES IN INCREASING ORDER
C        SECOND CARD  $$$$  FLYER LENGTH, WIDTH, THICKNESS, INITIAL SPACING,
C        FLYER DENSITY, FLYER MATERIAL (H FORMAT)
C        THIRD CARD  $$$$  TOTAL R, TOTAL L, TOTAL C, TOTAL TIME
C        FOURTH CARD   $$$$   FLYER DL/DX, MAXIMUM DISPLACEMENT
C
      COMMON TYPE,XLIM
      CALL APERTUR
      CALL REFSYM
      READ 210, (V(I),I=1,4)
  210 FORMAT (4E10.4)
      READ 211,AL,BW,TH,XN,RHO,TYPE
  211 FORMAT (5E10.4,A10)
      READ 212,R,L,CAP,TMAX
  212 FORMAT (4E10.4)
      READ 214,K1,XLIM
  214 FORMAT (2E10.4)
      PI = 3.141592654
      MU = 4.*PI*1.E-7
      IF (K1.EQ.0.) K1 = MU*AL/BW
      K2 = L - XN*K1
      PRINT 40, L,TMAX
   40 FORMAT(*1THE INITIAL L =*,E12.3,5X,*TMAX =*,E12.3)
      PRINT 213, TYPE
```

```
213 FORMAT (*0THIS FLYER IS A *,A10,* FLAT FLYER*///)
    HMAX = TMAX/299.
    DELPR = HMAX
    HMIN = HMAX/1000.
    H = HMIN
    AREA = AL*BW
    M = TH*AREA*RHO
    K3 = K1/(2.*M)
    DO 30 I=1,4
    PRINT 10,R,CAP,K1,K2,K3,V(I)
 10 FORMAT    (*0R =*,E12.3,4X,*CAP =*,E12.3,4X,*K1 =*,E12.3,4X,
   1 *K2 =*,E12.3,4X,*K3 =*,E12.3,4X,*V =*,E12.3 )
    PRINT 20,XN,RHO,AL,BW,TH,M
 20 FORMAT(*0XN =*,E12.3,4X,*RHO =*,E12.3,4X,*AL =*,E12.3,4X,
   1 *BW =*,E12.3,4X,*TH =*,E12.3,4X,*M =*,E12.3//)
    CALL CONTROL (I)
    EN = V(4)*V(4)*CAP/2.
 30 CONTINUE
    CALL MAP(EN)
    CALL OUTPUT(AL,BW,TH,EN,XN)
100 CALL ENDPLOT
    STOP
    END
```

```
      SUBROUTINE CONTROL (IND)
      REAL M,MU
      REAL L,K1,K2,K3
      COMMON /VAR/ R,CAP,K1,K2,K3,  L,XN,AREA,M,MU,PI
      COMMON /RKG/ YD(5),Y(5),XK(5,4),Q(5)
      COMMON /OUT/ TMAT(300),XMAT(300,4),CMAT(300,4),PMAT(300,4),V(4)
      COMMON /NUTS/ H,HMIN,HMAX,XLIML,INTCON,N,TMAX,DELPR,XLIMR,NPL
      DIMENSION YS(5),QS(5)
C
C   CMAT IN MEGAMPS
C    TMAT IN MICROSECONDS
C      XMAT IN MM
C     PMAT  (VELOCITY IN CM/USEC)
C
C
C  Y(1) = IDOT
C  Y(2) = XDOT
C  Y(3) = I
C  Y(4) = X
C  Y(5) = TIME
C
C
C  X(0) = XN
C  XDOT(0) = 0
C  IDOT(0) = V/(K1*XN+K2)
C   I(0) = 0
C
      XLIMR = 5.
      XLIML = 1.
      Y(1) = V(IND)/(K1*XN+K2)
      Y(2) = 0.
      Y(3) = 0.
      Y(4) = XN
      Y(5) = 0.
      TMAT (1) = Y(5)*1.E+6
      XMAT(1,IND) = 1.E+3 *Y(4)
      CMAT (1,IND) = Y(3)*1.E-6
      PMAT(1,IND) = Y(2)*1.E-4
```

```
      NPL = 1
      NIN = 4
      IEND = 1
      ICON = 1
      N = -NIN
      NX = NIN + 1
      TIME = DELPR
   20 CALL RKG
      GO TO (25,34,34),ICON
   25 TEST = TMAX - Y(NX)
      IF (TEST) 30,50,55
   30 HS = H
      H = TEST
      INTCON = 1
      ICON = 2
      IEND = 2
      GO TO 20
   34 H = HS
   35 CONTINUE
      NPL = NPL + 1
      TMAT(NPL) = Y(5)*1.E+6
      XMAT(NPL,IND) = 1.E+3*Y(4)
      CMAT(NPL,IND) = Y(3)*1.E-6
      PMAT(NPL,IND) = Y(2)*1.E-4
      GO TO (70,45) , IEND
   45 PRINT 15
   15 FORMAT (1H0)
      RETURN
C
   50 IEND = 2
      GO TO 35
   55 TEST = TIME - Y(NX)
      IF (TEST) 60,60,20
   60 DO 65 I=1,NX
      YS(I) = Y(I)
      QS(I) = Q(I)
   65 CONTINUE
      INTS = INTCON
      HS = H
```

```
      H = TEST
      ICON = 3
      INTCON = 1
      GO TO 20
   70 DO 75 I=1,NX
      Y(I) = YS(I)
      Q(I) = QS(I)
   75 CONTINUE
      H = HS
      INTCON = INTS
      ICON = 1
   80 TIME = TIME + DELPR
      GO TO 20
      END
```

```
      SUBROUTINE RKG
      DIMENSION A(4),B(4),C(4)
      COMMON /OUT/ TMAT(300),XMAT(300,4),CMAT(300,4),PMAT(300,4),V(4)
      COMMON /NUTS/ H,HMIN,HMAX,XLIML,INTCON,N,TMAX,DELPR,XLIMR,NPL
      COMMON /RKG/ YD(5),Y(5),XK(5,4),Q(5)
      COMMON /VAR/ R,CAP,K1,K2,K3,  L,XN,AREA,M,MU,PI
C     *
C       N        - ORDER OF SYSTEM TO BE SOLVED (NEGATIVE ON FIRST CALL)
C       Y(I)     - FUNCTION VALUES - TIME IS AUTOMATICALLY IN Y(N+1)
C       YD(I)    - FUNCTION DERIVATIVES
C       XK(I)    - STORAGE FOR INTERVAL CONTROL INFORMATION
C       H        - STEP SIZE
C       HMIN     - MINIMUM PERMITTED STEP SIZE
C       HMAX     - MAXIMUN PERMITTED STEP SIZE
C       XLIML    - MINIMUM TOLERANCE FOR DOUBLING (PERCENTAGE)
C       XLIMR    - MAXIMUM TOLERANCE FOR DOUBLING (PERCENTAGE)
C     *
C       INITIALIZATION PHASE WHEN N IS NEGATIVE OR ZERO.
C     *
      IF (N) 1, 1, 10
    1 NX = 1 -  N
      A(1) = .5
      A(2) = 1. - SQRT(.5)
      A(3) = 1. + SQRT(.5)
      A(4) = 1./6.
      B(1) = 2.
      B(2) = 1.
      B(3) = B(2)
      B(4) = B(1)
      C(1) = A(1)
      C(2) = A(2)
      C(3) = A(3)
      C(4) = C(1)
      N = - N
      YD(NX) = 1.0
      DO 5 I = 1, NX
      Q(I) = 0.0
    5 CONTINUE
      INTCON = 1
```

```
C     *
C       ENTER HERE WITH N POSITIVE AND CONTINUE INTEGRATION
C     *
C       TEST FOR POSSIBLE INTERVAL MODIFICATION
C     *
C       INTCON = 1   - NORMAL, DO NOT CHANGE H
C       INTCON = 2   - DO NOT PERMIT DOUBLING, SET INTCON TO 1
C       INTCON = 3   - HALVE H, SET INTCON = 2
C       INTCON = 4   - DOUBLE H, SET INTCON = 1
C     *
   10 GO TO (50, 20, 15, 25), INTCON
C     *
C       HALVE H
C     *
   15 H = 0.5 * H
      INTCON = 2
      GO TO 30
C     *
C       RESET NO DOUBLING INTERLOCK
C     *
   20 INTCON = 1
      GO TO 30
C     *
C       DOUBLE H
C     *
   25 H = H + H
      INTCON = 1
C     *
C       TEST H FOR BEING BETWEEN HMIN AND HMAX
C     *
   30 IF (H - HMIN) 35, 50, 40
   35 H = HMIN
      GO TO 50
   40 IF (HMAX - H) 45, 50, 50
   45 H = HMAX
C     *
C         INTEGRATE ONE TIME STEP
```

```
C     *
   50 IF (H ) 52, 90, 52
   52 DO 60 J = 1, 4
      CALL DERV
      DO 55 I = 1, NX
      TEMP = A(J) * (YD(I) - B(J) * Q(I))
      W = Y(I)
      Y(I) = Y(I) + H * TEMP
      TEMP = (Y(I) - W) / H
      Q(I) = Q(I) + 3.0 * TEMP - C(J) * YD(I)
      XK(I,J) = YD(I)
   55 CONTINUE
   60 CONTINUE
C     *
C       TEST RESULTS FOR INTERVAL MODIFICATION
C     *
      DO 85 I = 1, N
      TEMP = ABSF(XK(I,2) - XK(I,1))
      XL = XLIML * TEMP * 0.01
      XR = XLIMR * TEMP * 0.01
      TEMP = ABSF(XK(I,3) - XK(I,2))
      IF (XR - TEMP) 65, 70, 70
   65 INTCON = 3
      GO TO 90
   70 GO TO (75, 85, 90, 85), INTCON
   75 IF (TEMP - XL) 80, 85, 85
   80 INTCON = 4
   85 CONTINUE
   90 RETURN
      END
```

```
SUBROUTINE DERV
REAL M,MU
COMMON /OUT/ TMAT(300),XMAT(300,4),CMAT(300,4),PMAT(300,4),V(4)
COMMON /NUTS/ H,HMIN,HMAX,XLIML,INTCON,N,TMAX,DELPR,XLIMR,NPL
COMMON /VAR/ R,CAP,K1,K2,K3,  L,XN,AREA,M,MU,PI
REAL L,K1,K2,K3
COMMON /RKG/ YD(5),Y(5),XK(5,4),Q(5)
TEMP = -R*Y(1) - Y(3)/CAP -K1*K3*Y(3)**3 - 2.*K1*Y(1)*Y(2)
YD(1) = TEMP/(K1*Y(4) + K2)
YD(2) = K3*Y(3)*Y(3)
YD(3) = Y(1)
YD(4) = Y(2)
RETURN
END
```

```
      SUBROUTINE MAP (EN)
      COMMON /OUT/ TMAT(300),XMAT(300,4),CMAT(300,4),PMAT(300,4),V(4)
      COMMON /NUTS/ H,HMIN,HMAX,XLIML,INTCON,N,TMAX,DELPR,XLIMR,NPL
      COMMON /VAR/ R,CAP,K1,K2,K3,  L,XN,AREA,M,MU,PI
      REAL L,K1,K2,K3,MU,M
      REAL LT,KE
C
C   THE NEW X IS MEASURED FROM THE INITIAL SPACING OF THE FLYER
C
      PRINT 5,EN
    5 FORMAT (*0THE INITIAL STORED ENERGY =*,E12.3,* JOULES*)
      DO 20 J=1,4
      PRINT 10,V(J)
   10 FORMAT (*1*,10X,*VOLTAGE =*,E12.3///
     1 *0 I *,12X,*X (MM)*,14X,*TIME (US)*,9X,*IMPULSE (TAPS)*,8X,
     2 *V (CM/USEC)*,8X,*INDUCTANCE (UH)*,6X,*K.E. (JOULES)*///)
      K = 0
      DO 25 I=1,300
      K = K + 1
      KE = .5*M*(PMAT(I,J)*1.E+4)**2
      LT = (K1*XMAT(I,J)*1.E-3 + K2)*1.E+6
      P = PMAT(I,J)*10.*1.E+4*M/AREA
      X = XMAT(I,J) - XN*1.E+3
      PRINT 15,I,X,TMAT(I),P,PMAT(I,J),LT,KE
   15 FORMAT (* *,I3,6E20.3)
      IF (I.GE.300) GO TO 25
      IF (K.EQ.50) PRINT 10,V(J)
      IF (K.EQ.50) K = 0
   25 CONTINUE
   20 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE OUTPUT (AL,BW,TH,EN,XN)
      COMMON TYPE,XLIM
      COMMON /OUT/ TMAT(300),XMAT(300,4),CMAT(300,4),PMAT(300,4),V(4)
      COMMON /NUTS/ H,HMIN,HMAX,XLIML,INTCON,N,TMAX,DELPR,XLIMR,NPL
      DIMENSION A1(2),A2(2),A3(2),A4(2),A5(16)
      DATA((A1(I),I=1,2) = 20HDISPLACEMENT (MM)               )
      DATA((A2(I),I=1,2) = 20HVELOCITY (CM/(SEC)      )
      DATA((A3(I),I=1,2) = 20H    TIME ((S)                   )
      DATA((A4(I),I=1,2) = 20H CURRENT (MEGAMPS)              )
      DIMENSION IV(4)
      DIMENSION NPMAT(4)
      COMMON/ PLOTTER/ F(75)
      REAL MAX
C
C        THIS SUBROUTINE PROVIDES PLOTTED OUTPUT OF MATRIXED VALUES
C
C    F(1) = XMIN OF GRID
C    F(2) = YMIN OF GRID
C    F(3) = XMAX OF GRID
C    F(4) = YMAX OF GRID
      DO 7 I=1,4
      IV(I) = V(I)*1.E-3 + .5
    7 CONTINUE
      ENCODE(80,41,A5(9)) (IV(I),I=1,4)
   41 FORMAT (20H   VOLTAGE CODING     ,I3,8HKV (1), ,I3,8HKV (2), ,I3,
     1 8HKV (3), ,I3,8HKV (4),    ,16X)
      DO 85 J1=1,4
      DO 80 I=1,300
      XMAT(I,J1) = XMAT(I,J1) - XN*1.E+3
   80 CONTINUE
   85 CONTINUE
      IE = EN*1.E-3 + .5
      ENCODE (80,40,A5) IE,AL,BW,TH,TYPE
   40 FORMAT (I3,11HKJ BANK FOR,E11.3,4H M X,E11.3,4H M X,E11.3,8H M FLA
     1T ,A10,7H FLYERS)
      NP1 = 300*4
      NP2 = 300
      J = 20
```

```
      CALL AUXY(2)
      DO 10 I=1,4
      ISYM = 59 + I
      K = NP2*I -NP2 + 1
      CALL SEARCH (I,NP)
      CALL MINMAX (XMAT(K),NP,XMIN,XMAX,1)
      CALL MINMAX (CMAT(K),NP,CMIN,CMAX,1)
      CALL MINMAX (PMAT(K),NP,PMIN,PMAX,1)
      XMIN = 0.
      CMIN = -3.*CMAX
      PMIN = 0.
      CALL MINMAX (TMAT,NP,TMIN, MAX,1)
C
C      LARGEST X,P,C MATRICIES MUST BE LAST CALLED
C
      NPMAT(I) = NP
      CALL QUICK (2,TMAT,PMAT(K),NP ,ISYM,A3,J,A2,J,TMIN, MAX,PMIN,PMAX)
      F2 = F(4) + .14
      F1 = F(4) + .42
      CALL SYMBOL (F(1),F1,.14,A5,0.,80)
      CALL SYMBOL (F(1),F2,.14,A5(9),0.,80)
      ISYM = 56                                                         X
      CALL AUXML (1,1,TMAT,XMAT(K),NP ,ISYM,A1,J,XMIN,XMAX)
      ISYM = 35                                                         C
      CALL AUXML (2,1,TMAT,CMAT(K),NP ,ISYM,A4,J,CMIN,CMAX)
   10 CONTINUE
      CALL AUXY(0)
      CALL MINMAX (PMAT,NP1,P1,P2,1)
      DO 30 I=1,4
      L = 0
      IF(I.EQ.1) L=2
      ISYM = 59 + I
      K = NP2*I -NP2 + 1
      CALL QUICK (L,TMAT,PMAT(K),300,ISYM,A3,J,A2,J,TMAT(1),TMAT(300),
     1 P1,P2)
   30 CONTINUE
      CALL SYMBOL (F(1),F2,.14,A5(9),0.,80)
      CALL SYMBOL (F(1),F1,.14,A5,0.,80)
      DO 20 I=1,4
      L = 0
```

```
      IF (I.EQ.1) L = 2
      K = NP2*I -NP2 + 1
      ISYM = 59 + I
      CALL QUICK (L,XMAT(K),PMAT(K),NPMAT(I),ISYM,A1,J,A2,J,XMIN,XMAX,
     1 PMIN,PMAX)
   20 CONTINUE
      CALL SYMBOL (F(1),F1,.14,A5,0.,80)
      CALL SYMBOL (F(1),F2,.14,A5(9),0.,80)
      PRINT 75
   75 FORMAT (1H1)
      RETURN
      END
```

```
      SUBROUTINE SEARCH (I,NP)
      COMMON /OUT/ TMAT(300),XMAT(300,4),CMAT(300,4),PMAT(300,4),V(4)
      COMMON /VAR/ R,CAP,K1,K2,K3,  L,XN,AREA,M,MU,PI
      COMMON TYPE,XLIM
C     XLIMIT AND XMAT IN MM
      XLIMIT = XLIM*1.E+3
      DO 10 K=1,300
      TEST = XMAT(K,I)  - XLIMIT
      NP = K
      IF(TEST.GT.0.) GO TO 20
   10 CONTINUE
   20 RETURN
      END
```

DISTRIBUTION:

Don Keller
General Research Corp.,
Santa Barbara, Calif.
Attn: M. E. Graham

T. F. Meagher
Kaman Nuclear,
Colorado Springs, Colo.

CDR. L. J. Ratto
LMSC,
Sunnyvale, Calif.
Attn: R. R. Taylor

L. W. Woodruff, LRL, L125
Q. C. Johnson, LRL, L404
Technical Information Division, LRL, L3
T. B. Lane, 1540
R. G. Clem, 1730
S. J. Buchsbaum, 5000
O. E. Jones, 5130
L. W. Davison, 5133
W. Herrmann, 5160
D. E. Munson, 5163; Attn: R. P. Reed
A. W. Snyder, 5220
E. H. Beckner, 5240
M. Cowan, 5261
L. M. Berry, 5430
L. E. Lamkin, 7300
W. T. Moffat, 7330
M. McWhirter, 7340
F. Mathews, 7342
J. D. Patrick, 7335; Attn: W. K. Tucker
T. B. Cook, Jr., 8000
L. Gutierrez, 8100
J. W. Pearce, 8120
J. D. Benton, 8121
R. S. Jacobson, 8122 (2)
W. D. Zinke, 8122
D. J. Bohrer, 8123
C. M. Potthoff, 8124
R. M. Hargreaves, 8125
D. E. Gregson, 8130
L. E. Davies, 8150
G. L. Scholer, 8153
R. D. Cozine, 8154

J. S. Anderson, 8155
R. A. Baroody, 8160
G. E. Brandvold, 8170
P. D. Gildea, 8174
J. C. King, 8300
J. M. Brierly, 8311
D. R. Adolphson, 8312
J. W. Weihe, 8320
G. W. Anderson, Jr., 8330
A. R. Willis, 8335
C. M. Robinson, Jr., 8336
C. S. Hoyle, 8337
J. L. Wirth, 8340
J. A. Mogford, 8341; Attn: K. Marx
H. R. Sheppard, 8342
S. G. Cain, 8343
R. W. Reynolds, 8343 (6)
AEC/DTI, Oak Ridge, Tenn. (3)
Technical Information Division III, 3413
Acquisitions and Cataloging Division, 3421
Subsequent Distribution and Records Depository, 3428-2 (3)
M. J. Connors, 8216
D. L. Rasmussen, 8231/Central Technical File, 8232-2
B. F. Hefley/E. Bodie, 8232
Central Technical File, 8232-2 (3)

Subsequent Distribution: 1/19/70 -- Central Technical File, 8232-2 (25)